NAG Fortran Library Routine Document

F08BHF (DTZRZF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F08BHF (DTZRZF) reduces the m by n ($m \leq n$) real upper trapezoidal matrix A to upper triangular form by means of orthogonal transformations.

2 Specification

```
SUBROUTINE F08BHF (M, N, A, LDA, TAU, WORK, LWORK, INFO)
INTEGER             M, N, LDA, LWORK, INFO
double precision    A(LDA,*), TAU(*), WORK(*)
```

The routine may be called by its LAPACK name ***dtzrzf***.

3 Description

The m by n ($m \leq n$) real upper trapezoidal matrix A given by

$$A = (R_1 \quad R_2),$$

where R_1 is an m by m upper triangular matrix and R_2 is an m by $(n-m)$ matrix, is factorized as

$$A = (R \quad 0)Z,$$

where R is also an m by m upper triangular matrix and Z is an n by n orthogonal matrix.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia URL: http://www.netlib.org/lapack/lug

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq 0$.

3: A(LDA,$*$) – ***double precision*** array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the leading M by N upper trapezoidal part of the array A must contain the matrix to be factorized.

On exit: the leading M by M upper triangular part of A contains the upper triangular matrix R, and elements $\mathrm{M}+1$ to N of the first M rows of A, with the array TAU, represent the orthogonal matrix Z as a product of m elementary reflectors.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F08BHF (DTZRZF) is called.

Constraint: LDA $\geq \max(1, \mathrm{M})$.

5: TAU$(*)$ – ***double precision*** array *Output*

Note: the dimension of the array TAU must be at least $\max(1, \mathrm{M})$.

On exit: the scalar factors of the elementary reflectors.

6: WORK$(*)$ – ***double precision*** array *Workspace*

Note: the dimension of the array WORK must be at least $\max(1, \mathrm{LWORK})$.

On exit: if INFO $= 0$, WORK(1) contains the minimum value of LWORK required for optimal performance.

7: LWORK – INTEGER *Input*

On entry: the dimension of the array WORK as declared in the (sub)program from which F08BHF (DTZRZF) is called.

If LWORK $= -1$, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK $\geq \mathrm{M} \times nb$, where nb is the optimal ***block size***.

Constraint: LWORK $\geq \max(1, \mathrm{M})$ or LWORK $= -1$.

8: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed factorization is the exact factorization of a nearby matrix $A + E$, where

$$\|E\|_2 = O\epsilon\|A\|_2$$

and ϵ is the ***machine precision***.

8 Further Comments

The total number of floating point operations is approximately $4m^2(n - m)$.

The complex analogue of this routine is F08BVF (ZTZRZF).

9 Example

This example solves the linear least squares problems

$$\min_x \left\|b_j - Ax_j\right\|_2, \quad j = 1, 2$$

for the minimum norm solutions x_1 and x_2, where b_j is the jth column of the matrix B,

$$A = \begin{pmatrix} -0.09 & 0.14 & -0.46 & 0.68 & 1.29 \\ -1.56 & 0.20 & 0.29 & 1.09 & 0.51 \\ -1.48 & -0.43 & 0.89 & -0.71 & -0.96 \\ -1.09 & 0.84 & 0.77 & 2.11 & -1.27 \\ 0.08 & 0.55 & -1.13 & 0.14 & 1.74 \\ -1.59 & -0.72 & 1.06 & 1.24 & 0.34 \end{pmatrix} \quad \text{and} \quad B = \begin{pmatrix} 7.4 & 2.7 \\ 4.2 & -3.0 \\ -8.3 & -9.6 \\ 1.8 & 1.1 \\ 8.6 & 4.0 \\ 2.1 & -5.7 \end{pmatrix}.$$

The solution is obtained by first obtaining a QR factorization with column pivoting of the matrix A, and then the RZ factorization of the leading k by k part of R is computed, where k is the estmiated rank of A. A tolerance of 0.01 is used to estimate the rank of A from the upper triangular factor, R.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

9.1 Program Text

```
*     F08BHF Example Program Text
*     Mark 21 Release. NAG Copyright 2004.
*     .. Parameters ..
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
      INTEGER          MMAX, NB, NMAX, NRHSMX
      PARAMETER        (MMAX=8,NB=64,NMAX=8,NRHSMX=2)
      INTEGER          LDA, LDB, LWORK
      PARAMETER        (LDA=MMAX,LDB=MMAX,LWORK=2*NMAX+(NMAX+1)*NB)
      DOUBLE PRECISION ONE, ZERO
      PARAMETER        (ONE=1.0D0,ZERO=0.0D0)
*     .. Local Scalars ..
      DOUBLE PRECISION TOL
      INTEGER          I, IFAIL, INFO, J, K, M, N, NRHS
*     .. Local Arrays ..
      DOUBLE PRECISION A(LDA,NMAX), B(LDB,NRHSMX), RNORM(NMAX),
     +                 TAU(NMAX), WORK(LWORK)
      INTEGER          JPVT(NMAX)
*     .. External Functions ..
      DOUBLE PRECISION DNRM2
      EXTERNAL         DNRM2
*     .. External Subroutines ..
      EXTERNAL         DCOPY, DGEQP3, DORMQR, DORMRZ, DTRSM, DTZRZF,
     +                 F06DBF, F06QHF, X04CAF
*     .. Intrinsic Functions ..
      INTRINSIC        ABS
*     .. Executable Statements ..
      WRITE (NOUT,*) 'F08BHF Example Program Results'
      WRITE (NOUT,*)
*     Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) M, N, NRHS
      IF (M.LE.MMAX .AND. N.LE.NMAX .AND. M.GE.N .AND. NRHS.LE.NRHSMX)
     +    THEN
*
*        Read A and B from data file
*
         READ (NIN,*) ((A(I,J),J=1,N),I=1,M)
         READ (NIN,*) ((B(I,J),J=1,NRHS),I=1,M)
*
*        Initialize JPVT to be zero so that all columns are free
*
         CALL F06DBF(N,0,JPVT,1)
*
*        Compute the QR factorization of A with column pivoting as
*        A = Q*(R11 R12)*(P**T)
*              ( 0  R22)
*
         CALL DGEQP3(M,N,A,LDA,JPVT,TAU,WORK,LWORK,INFO)
*
```

```
*        Compute C = (C1) = (Q**T)*B, storing the result in B
*                    (C2)
*
         CALL DORMQR('Left','Transpose',M,NRHS,N,A,LDA,TAU,B,LDB,WORK,
     +               LWORK,INFO)
*
*        Choose TOL to reflect the relative accuracy of the input data
*
         TOL = 0.01D0
*
*        Determine and print the rank, K, of R relative to TOL
*
         DO 20 K = 1, N
            IF (ABS(A(K,K)).LE.TOL*ABS(A(1,1))) GO TO 40
   20    CONTINUE
   40    K = K - 1
*
         WRITE (NOUT,*) 'Tolerance used to estimate the rank of A'
         WRITE (NOUT,99999) TOL
         WRITE (NOUT,*) 'Estimated rank of A'
         WRITE (NOUT,99998) K
         WRITE (NOUT,*)
*
*       Compute the RZ factorization of the K by K part of R as
*       (R11 R12) = (T 0)*Z
*
         CALL DTZRZF(K,N,A,LDA,TAU,WORK,LWORK,INFO)
*
*        Compute least-squares solutions of triangular problems by
*        back substitution in T*Y1 = C1, storing the result in B
*
         CALL DTRSM('Left','Upper','No transpose','Non-Unit',K,NRHS,ONE,
     +              A,LDA,B,LDB)
*
*        Compute estimates of the square roots of the residual sums of
*        squares (2-norm of each of the columns of C2)
*
         DO 60 J = 1, NRHS
            RNORM(J) = DNRM2(M-K,B(K+1,J),1)
   60    CONTINUE
*
*        Set the remaining elements of the solutions to zero (to give
*        the minimum-norm solutions), Y2 = 0
*
         CALL F06QHF('General',N-K,NRHS,ZERO,ZERO,B(K+1,1),LDB)
*
*        Form W = (Z**T)*Y
*
         CALL DORMRZ('Left','Transpose',N,NRHS,K,N-K,A,LDA,TAU,B,LDB,
     +               WORK,LWORK,INFO)
*
*        Permute the least-squares solutions stored in B to give X = P*W
*
         DO 100 J = 1, NRHS
            DO 80 I = 1, N
               WORK(JPVT(I)) = B(I,J)
   80       CONTINUE
            CALL DCOPY(N,WORK,1,B(1,J),1)
  100    CONTINUE
*
*        Print least-squares solutions
*
         IFAIL = 0
         CALL X04CAF('General',' ',N,NRHS,B,LDB,
     +               'Least-squares solution(s)',IFAIL)
*
*        Print the square roots of the residual sums of squares
*
         WRITE (NOUT,*)
         WRITE (NOUT,*)
     +     'Square root(s) of the residual sum(s) of squares'
```

```
         WRITE (NOUT,99999) (RNORM(J),J=1,NRHS)
      ELSE
         WRITE (NOUT,*)
     +     'One or more of MMAX, NMAX and NRHSMX is too small, ',
     +     'and/or M.LT.N'
      END IF
      STOP
*
99999 FORMAT (5X,1P,6E11.2)
99998 FORMAT (1X,I8)
      END
```

9.2 Program Data

```
F08BHF Example Program Data

  6  5  2                          :Values of M, N and NRHS

 -0.09   0.14  -0.46   0.68   1.29
 -1.56   0.20   0.29   1.09   0.51
 -1.48  -0.43   0.89  -0.71  -0.96
 -1.09   0.84   0.77   2.11  -1.27
  0.08   0.55  -1.13   0.14   1.74
 -1.59  -0.72   1.06   1.24   0.34 :End of matrix A

  7.4    2.7
  4.2   -3.0
 -8.3   -9.6
  1.8    1.1
  8.6    4.0
  2.1   -5.7                        :End of matrix B
```

9.3 Program Results

```
 F08BHF Example Program Results

 Tolerance used to estimate the rank of A
        1.00E-02
 Estimated rank of A
        4

 Least-squares solution(s)
             1          2
 1      0.6344     3.6258
 2      0.9699     1.8284
 3     -1.4402    -1.6416
 4      3.3678     2.4307
 5      3.3992     0.2818

 Square root(s) of the residual sum(s) of squares
        2.54E-02   3.65E-02
```